




**PML Library:
PD4 - Warfarin IV-bolus
Turnover Model with Tlag**

PD4: Background


- Warfarin is often prescribed as an anticoagulant.
- Warfarin acts indirectly to prevent blood clotting, by inhibiting the production of reduced Vitamin K, that is involved in the formation of the Prothrombin Complex, as shown below:



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD4: Protocol

- An IV bolus dose of Warfarin is administered
- Response is measured as % of Prothrombin Complex Activity (PCA) compared to normal (100%)
- % PCA vs Time:



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD4: Warfarin PK parameters

- Warfarin PK is well characterized from previous studies.
- A 1-compartment bolus model with micro constants (V , K_e) will be used for the PK parameters in the following equation:
- $C_w = \text{Dose}/V * \exp(-K_e * t)$
 - Dose = 1.05
 - $V = 1$
 - $K_e = 0.0228$

PD4: Objectives

- Use indirect response to model Warfarin-PCA interaction
- Elaborate on a model with and without lag time in the dynamics
- Compare the two models by AIC and parameter precision
- Optional: extend to a precursor pool model


PD4: Derivation of Warfarin Inhibitory function

- The empirically found inhibitory function of warfarin on the synthesis rate of PCA (assuming $I_{\max} = 1$) is:
- $I(C) = 1 - C^Y / (IC_{50}^Y + C^Y)$
- Substitute $(IC_{50}^Y + C^Y) / (IC_{50}^Y + C^Y)$ for 1:
- $= (IC_{50}^Y + C^Y) / (IC_{50}^Y + C^Y) - C^Y / (IC_{50}^Y + C^Y)$
- $= IC_{50}^Y / (IC_{50}^Y + C^Y)$
- Divide numerator and denominator by IC_{50}^Y gives:
- $= 1 / (1 + C^Y / IC_{50}^Y)$
- $I(C) = 1 / (1 + (C / IC_{50})^Y)$

PD4: Base Turnover Model

- The turnover rate of PCA is adapted from the base inhibitory turnover model:
- $dPCA/dt = k_{in} * I(C_w) - k_{out} * PCA$
 - K_{in} = rate of production
 - K_{out} = rate of loss
- At baseline, $R_0 = k_{in}/k_{out}$, therefore $k_{in} = R_0 * k_{out}$
- Set $R_0 = P_0$, and substitute $P_0 * k_{out}$ for k_{in} :
- $dPCA/dt = P_0 * k_{out} * I(C_w) - k_{out} * PCA$
- Rearranges to:
- $dPCA/dt = k_{out} * (P_0 * I(C_w) - PCA)$


PD4: k_{out} Initial Estimate



$$k_{out} = - \frac{\ln(124.4) - \ln(56.77)}{0 - 24} \approx 0.03 \text{ h}^{-1}$$

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD4: IC₅₀ Initial Estimate



- IC₅₀ estimate: solve C_w at time = 80:
- $C_w = \text{Dose}/V * \exp(-K_e * t)$
- $IC_{50} = 1.05/1 * \exp(-0.0228 * 80) = 0.16$

PD4: Summary of Initial Estimates

- P_0 is baseline PCA activity = 124
- $K_{out} = 0.03$, derived from onset slope in exploratory plot
- IC_{50} = warfarin concentration when response recovers by 50% = 0.16
- n = gamma exponent, can be found by fitting built-in inhibitory indirect response model = 4

PD4: Built-in PK/indirect: inhibition of production

☐ Population? | Structure | Parameters | Input Options | Initial Estimates | Run Options | Model Text | Plots | no warnings

Type: PK/Indirect Set WNL Model

Parameterization: Absorption: Num Compartments: Parameters: Statements:

Micro Intravenous 1 V $\text{deriv}(A1 = -K_e * A1)$

☐ tlag? ☐ Elim. Cpt.? Ke $\text{deriv}(E = K_{in} * (1 - I_{max} * C^{\text{gam}} / (C^{\text{gam}} + IC50^{\text{gam}})) -$

☐ Closed form? ☒ Freeze PK? K_{in} $\text{dosepoint}(A1)$

☐ Infusions possible? ☒ Freeze PK? K_{out} $C = A1 / V$


☐ Effect Cpt? Indirect: Inhib. Limited OF Build-up I_{max} $\text{sequence}\{E = K_{in} / K_{out}\}$

Exponent ☐ Freeze Indirect? IC50 $\text{error}(EEps = 0.493378)$

gam $\text{observe}(EObs = E + EEps)$

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD4: Graphical Model



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD4: Base Turnover Model 1: PML Code

```
1 test(){
2     #dose delivered to central compartment
3     dosepoint(A1)
4     #initial concentration of warfarin
5     Cw = A1 / V
6     #differential equation for PK model
7     deriv(A1 = -Ke * A1)
8     #inhibitory function of warfarin
9     Inh = 1 / (1 + (Cw/IC50)^n )
10    #initial value for PCA set to P0 (baseline parameter)
11    sequence{PCA=P0;}
12    #differential equation for PCA
13    deriv(PCA=Kout*(P0*Inh - PCA))
14    #observed PCA with additive error term
15    error(Eps=1)
16    observe(PCAObs = PCA+Eps)
17    #PK parameters as frozen fixed effects
18    fixef(V(freeze) = c(, 1, ))
19    fixef(Ke(freeze) = c(, 0.0228, ))
20    #turnover model parameters as fixed effects with initial estimates
21    fixef(IC50 = c(, .16, ))
22    fixef(n = c(, 4, ))
23    fixef(Kout = c(, .03, ))
24    fixef(P0 = c(, 124, ))
25 }
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD4: Base Turnover Model 2 with Lag Time: PML Code

```
1 test(){
2     #dose delivered to central compartment
3     dosepoint(A1)
4     #initial warfarin concentration in central compartment
5     Cw = A1 / V
6     #differential equation for PK model
7     deriv(A1 = -Ke * A1)
8     #inhibitory function of warfarin
9     Inh = 1 / (1 + (Cw/IC50)^n )
10    #time switch for turnover model function
11    double(flag)
12    #sequence block: initial PCA with on/off switch (flag)
13    sequence{
14        PCA=P0;
15        flag=0;
16        sleep(Tlag);
17        flag=1;
18    }
19    #differential equation for PCA
20    deriv(PCA = flag*(Kout*(P0*Inh - PCA)))
21    #observed PCA with additive error model
22    error(Eps=1)
23    observe(PCAObs = PCA+Eps)
24    #PK parameters as frozen fixed effects
25    fixef(V(freeze) = c(, 1, ))
26    fixef(Ke(freeze) = c(, 0.0228, ))
27    #turnover model parameters as fixed effects with intitial estimates
28    fixef(Tlag = c(, 1, ))
29    fixef(IC50 = c(, .16, ))
30    fixef(n = c(, 4, ))
31    fixef(Kout = c(, .03, ))
32    fixef(P0 = c(, 124, ))
33 }
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)




Demo

PD4: Summary

- Fix PK parameters from previous study
- Develop an indirect response model
- Derive initial estimates
- Fit turnover models with and without lagtime to the data
- Review results

PD4: Base Turnover Model Results

- Fit




- Final Parameter Estimates

	Parameter	Estimate	Units	Stderr	CV%	2.5% CI	97.5% CI	Var. Inf. factor
1	V	1		0	0	1	1	0
2	Ke	0.0228		0	0	0.0228	0.0228	0
3	IC50	0.261284		0.020685502	7.9168652	0.20811048	0.31445752	1.5086E-05
4	n	2.99115		1.2677539	42.383495	-0.26769958	6.2499996	0.053706
5	Kout	0.030771		0.0048800615	15.859288	0.018226463	0.043315537	8.8425E-07
6	P0	120.951		4.0632499	3.3594182	110.50613	131.39587	0.63028
7	stdev0	4.4175		0.98778034	22.360619	1.878342	6.956658	

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD4: Base Turnover Model with Lag Time Results

- Fit




- Final Parameter Estimates

	Parameter	Estimate	Units	Stderr	CV%	2.5% CI	97.5% CI	Var. Inf. factor
1	V	1		0	0	1	1	0
2	Ke	0.0228		0	0	0.0228	0.0228	0
3	Tlag	6.54598		1.0782857	16.472487	3.5521802	9.5397798	0.13878
4	IC50	0.235077		0.013841243	5.8879613	0.19664757	0.27350643	1.9882E-05
5	n	1.51175		0.20318711	13.440523	0.94761239	2.0758876	0.0039438
6	Kout	0.0601138		0.0086395602	14.372008	0.036126545	0.084101055	8.2777E-06
7	P0	122.223		2.9644392	2.4254349	113.9924	130.4536	0.92002
8	stdev0	2.93445		0.65616309	22.360684	1.11265	4.75625	

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)


PD4: Use Table function to view fitted lagtime

- In the model Run Options tab, set up a simulation table using the sequence statement.
- Use the result “Table01” to produce an overlay plot with more predicted points to view the lagtime



PD4: Develop a mechanistic precursor pool model


- Warfarin is often prescribed as an anticoagulant.
- Warfarin acts indirectly to prevent blood clotting, by inhibiting the production of reduced Vitamin K, that is involved in the formation of the Prothrombin Complex, as shown below:



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD4: Develop a mechanistic precursor pool model

- Models 1 & 2 are the basic turnover model both with and without a lag time.
- Warfarin acts one step upstream, inhibiting the formation of reduced Vitamin K, which mediates the formation of PCA from the precursor pool.
- Model 3 is a precursor pool model replacing the lag time with a transit compartment with its own exit rate constant k_1 .
- Model 4 is similar to 3, but k_1 is set equal to k_{out} .



Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

PD4: Precursor Pool Model 3: PML Code

```
1 test(){
2   #Warfarin PK
3   deriv(A1 = -Ke * A1)
4   dosepoint(A1)
5   Cw = A1 / V
6   #Inhibitory function of warfarin
7   Inh = 1 / (1 + (Cw/IC50)^n )
8   #sequence: initial conditions for PCA and its Precursor pool
9   sequence{Precur=P0}
10  sequence{PCA=P0}
11  #differential equations for PCA and Precursor pool
12  deriv(Precur=P0*Kout*Inh - K1*Precur)
13  deriv(PCA=K1*Precur - Kout*PCA)
14  #observed PCA with additive error model
15  error(Eps=1)
16  observe(PCAObs = PCA + Eps)
17  #PK parameters as frozen fixed effects
18  fixef(V(freeze) = c(, 1, ))
19  fixef(Ke(freeze) = c(, 0.0228, ))
20  #Precursor pool model parameters with initial estimates
21  fixef(IC50 = c(, .16, ))
22  fixef(n = c(, 4, ))
23  fixef(K1 = c(0, .06, ))
24  fixef(Kout = c(0, .06, ))
25  fixef(P0 = c(, 124, ))
26 }
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 3rd Edition, Swedish Pharmacology Press (2015)

PD4: Precursor Pool Model 4: PML Code

```
1 test(){
2   #Warfarin PK
3   dosepoint(A1)
4   deriv(A1 = -Ke * A1)
5   Cw = A1 / V
6   #inhibitory function of warfarin
7   Inh = 1 / (1 + (Cw/IC50)^n )
8   #set Kloss = Kout
9   Kloss=Kout
10  #Sequence: initial conditions for PCA and its precursor pool
11  sequence{Precur=P0}
12  sequence{PCA=P0}
13  #differential equations for PCA and precursor pool
14  deriv(Precur=P0*Kout*Inh - Kloss*Precur)
15  deriv(PCA=Kloss*Precur - Kout*PCA)
16  #observed PCA with additive error model
17  error(Eps=1)
18  observe(PCAObs = PCA+Eps)
19  #PK parameters as frozen fixed effects
20  fixef(V(freeze) = c(, 1, ))
21  fixef(Ke(freeze) = c(, 0.0228, ))
22  #Precursor pool Kloss eq Kout model parameters with initial estimates
23  fixef(IC50 = c(, .16, ))
24  fixef(n = c(, 4, ))
25  fixef(Kout = c(, .06, ))
26  fixef(P0 = c(, 124, ))
27 }
```

Gabrielsson & Weiner, Pharmacokinetic and Pharmacodynamic Data Analysis - Concepts and Applications, 5th Edition, Swedish Pharmacology Press (2015)

Questions?



PML School: Materials

- Each model will be made available in Certara Forum
 - Link to live webinar and presentation slides
 - <https://support.certara.com/forums/forum/34-pml-school/>
 - Model text as file download
 - Can be imported into Phoenix model object to be run on a new dataset
 - Questions and comments can be exchanged in the Forum
 - Or can be entered into the Certara Support portal at:
 - <https://support.certara.com/support>
 - Or can be sent as emails to support@certara.com
- PML School on Twitter: @PML_School
 - Notifications on updates of materials, Q&A and discussions
 - Announcements of new sessions
- PML School on Youtube:
 - <https://www.youtube.com/user/CertaraLP/videos>

- A wide range of On Demand and Classroom courses are available through Certara University:
 - Introductory, intermediate and advanced instruction in Phoenix WinNonlin, Population Modeling using NLME, IVIVC Toolkit
 - Fundamentals of Pharmacokinetics and Pharmacodynamics
 - Noncompartmental data analysis
 - Programming Bootcamp
 - Partner Lectures and Webinar series
- Please visit our [Certara University](#) web site for more information

Coming up...



Turnover Model 4: IV Infusions

Apply a turnover model to multiple IV dosing response data

June 22, 2017 | 10am EST

Presenter: Bernd Wendt

- Dr. Wendt will present PD7, which covers a turnover model, using stimulation of loss, at different dose levels.
- Register for future PML School webinars here:
- <https://www.certara.com/software/pkpd-modeling-and-simulation/pml-school/>